This is a really cool looking figure, showing a mountain like 3D figure with contour profiles, and a curve traced out on its surface. The curve has arrows indicating an orientation. However, I did not end up using this figure in the book.

```
<< peeters` ;
peeters`setGitDir[ "../project/figures/GAelectrodynamics" ]
```

```
/Users/pjoot/project/figures/GAelectrodynamics
```

```
(*https://mathematica.stackexchange.com/a/20575/10*)

n = 6;
SeedRandom[17];
λ = RandomReal[GammaDistribution[4, 1 / 5], n];
ϵ = RandomReal[GammaDistribution[3, 1 / 4], n];
x0 = RandomVariate[BinormalDistribution[{2.5, 2}, {2, 1}, 1 / 2], n];

f[x_, {λ_, ϵ_, x0_}] :=
  Sum[λ[[i]] Exp[-(Norm[x - x0[[i]]] / ϵ[[i]])^2], {i, 1, Length[λ]}];

dem = Plot3D[f[{x, y}, {λ, ϵ, x0}], {x, -1, 5}, {y, -1, 5},
   PlotRange → {Full, Full, Full}, PlotStyle → Opacity[0.9],
   MeshFunctions → {#3 &}, MeshStyle → GrayLevel[0.7]];

γ[t_] := {5 / 2, 2} + {9 Sin[4 t], -6 Cos[4 t]} / 5;
path[t_, {λ_, ϵ_, x0_}] := With[{x = γ[t]}, Append[x, f[x, {λ, ϵ, x0}]]];

hike = Module[{asz},
   asz = 0.05;
   ParametricPlot3D[
     path[t, {λ, ϵ, x0}], {t, 0, π / 2}, PlotStyle → Directive[
       Thick
       , Red
      ]
     , PlotRange → {{0, 5}, {0, 5}, {0, 2.5}}
    ] /. Line[pts_] ↦
     {Arrowheads[{asz, asz, asz, asz, asz, 0}], Arrow[Tube[pts, {0.02}], {0, .2}]}] ;


pl = Show[hike, dem, Boxed → False, BoxRatios → Automatic, AxesLabel → {x, y, z}]
```




```
peeters`exportForLatex["orientedSurfaceFig1", pl]
```

```
{orientedSurfaceFig1.eps, orientedSurfaceFig1pn.png}
```