```
<< CliffordBasic`;
$SetSignature = {3, 0};
Import[
 "https://raw.githubusercontent.com/jkuczm/MathematicaCellsToTeX/master/NoInstall.
   m"]
```

Reciprocal basis computation with conventional vector algebra.

Same calculation using bivectors

```
ClearAll[x1, x2, inverse]
x1 = e[1] + e[2]; x2 = e[1] + 2 e[2];
x12 = OuterProduct[ x1, x2];
inverse[a_] := a / GeometricProduct[a, a] ;
x12inverse = inverse[x12];
s1 = InnerProduct[ x2, x12inverse];
s2 = InnerProduct[ x1, -x12inverse];
s1
s2
dots[a_, b_] :=
  {a , "·", b , " = ", InnerProduct[a // ReleaseHold, b // ReleaseHold]};
MapThread[dots, {{x1 // HoldForm, x2 // HoldForm, x1 // HoldForm, x2 // HoldForm} ,
   {s1 // HoldForm, s1 // HoldForm, s2 // HoldForm, s2 // HoldForm}}] // Grid
```

```
2 e[1] - e[2]

-e[1] + e[2]

x1  ·  s1  =  1
x2  ·  s1  =  0
x1  ·  s2  =  0
x2  ·  s2  =  1
```

Initial rough calculations (reformatted for display above) : Problem 2.2.

```
ClearAll[x1, x2, inverse, reciprocalFrame, s1, s2]

inverse := # / GeometricProduct[#, #] & ;

reciprocalFrame[x1_, x2_] := Module[{ix12},
  ix12 = OuterProduct[ x1, x2] // inverse;

  { InnerProduct[ x2, ix12], InnerProduct[ x1, -ix12]}
 ]

x1 = e[1] + 2 e[2]; x2 = e[2] - e[3];
{s1, s2} = reciprocalFrame[x1, x2];
s1
s2
InnerProduct[#[[1]], #[[2]]] & /@ {{x1, s1}, {x2, s2}, {x1, s2}, {x2, s1}}
```

$$\frac{e[1]}{3} + \frac{e[2]}{3} + \frac{e[3]}{3}$$

$$-\frac{e[1]}{3} + \frac{e[2]}{6} - \frac{5\,e[3]}{6}$$

```
{1, 1, 0, 0}
```

Wolfgang seeing unexpected results for the following?

```
{e1, e2, e3} = {e[1], e[2], e[3]};
x1 = e[1] + 2 e[2]; x2 = e[2] - e[3];
s1 = 1 / 3 (e1 + e2 + e3) ;
s2 = 1 / 6 * (-2 * e1 + 1 * e2 - 5 * e3);
InnerProduct[#[[1]], #[[2]]] & /@ {{x1, s1}, {x2, s2}, {x1, s2}, {x2, s1}}
```

```
{1, 1, 0, 0}
```

Display the cells for latex